```
In[54]:= Off[General::"spell1"]
```

# A variable shift

This is just an example of *Mathematica* code for a matrix of a linear transformation in a space of polynomials. Here the transformation is a variable shift. Mathematical definition is $(T\ p)(x) = p(x+c)$. The command below will produce the matrix of this linear transformation relative to the standard basis for polynomials.

```
In[10]:= MatrixForm[mS = Transpose[Table[PadRight[CoefficientList[Collect[Expand[
              (IdentityMatrix[5][[k]].{1, x, x^2, x^3, x^4}) /. {x → (x + c)}
            ], x], x], 5], {k, 1, 5}]]]
```

Out[10]//MatrixForm=

$$\begin{pmatrix} 1 & c & c^2 & c^3 & c^4 \\ 0 & 1 & 2c & 3c^2 & 4c^3 \\ 0 & 0 & 1 & 3c & 6c^2 \\ 0 & 0 & 0 & 1 & 4c \\ 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

Let me explain the above command, starting from inside.

I like to think of polynomials as "dot products" of a list of coefficients and powers:

```
In[5]:= {a0, a1, a2, a3}.{1, x, x^2, x^3}
```

Out[5]= $a0 + a1\,x + a2\,x^2 + a3\,x^3$

This is the reason why I write the powers of the independent variable in such a strange way:

```
In[6]:= (IdentityMatrix[5][[3]].{1, x, x^2, x^3, x^4})
```

Out[6]= $x^2$

Above, `IdentityMatrix[5]` is the 5-by-5 identity matrix, , `IdentityMatrix[5][[3]]` is its third row.

```
In[7]:= IdentityMatrix[5]
```

```
Out[7]= {{1, 0, 0, 0, 0}, {0, 1, 0, 0, 0}, {0, 0, 1, 0, 0}, {0, 0, 0, 1, 0}, {0, 0, 0, 0, 1}}
```

```
In[8]:= IdentityMatrix[5][[3]]
```

```
Out[8]= {0, 0, 1, 0, 0}
```

The following part makes the substitution

```
In[3]:= (IdentityMatrix[5][[3]].{1, x, x^2, x^3, x^4}) /. {x → (x + c)}
```

Out[3]= $(c + x)^2$

```
In[9]:=  {a0, a1, a2, a3}.{1, x, x^2, x^3} /. {x → (x + c)^2}
```

Out[9]= $a0 + a1\,(c + x)^2 + a2\,(c + x)^4 + a3\,(c + x)^6$

Now try

```
In[12]:=  Expand[{a0, a1, a2, a3}.{1, x, x^2, x^3} /. {x → (x + c)^2}]
```

Out[12]= $a0 + a1\,c^2 + a2\,c^4 + a3\,c^6 + 2\,a1\,c\,x + 4\,a2\,c^3\,x + 6\,a3\,c^5\,x + a1\,x^2 + 6\,a2\,c^2\,x^2 + 15\,a3\,c^4\,x^2 + 4\,a2\,c\,x^3 + 20\,a3\,c^3\,x^3 + a2\,x^4 + 15\,a3\,c^2\,x^4 + 6\,a3\,c\,x^5 + a3\,x^6$

Now try, notice that I have to indicate the variable in the command below:

```
In[13]:=  Collect[Expand[{a0, a1, a2, a3}.{1, x, x^2, x^3} /. {x → (x + c)^2}], x]
```

Out[13]= $a0 + a1\,c^2 + a2\,c^4 + a3\,c^6 + (2\,a1\,c + 4\,a2\,c^3 + 6\,a3\,c^5)\,x + (a1 + 6\,a2\,c^2 + 15\,a3\,c^4)\,x^2 + (4\,a2\,c + 20\,a3\,c^3)\,x^3 + (a2 + 15\,a3\,c^2)\,x^4 + 6\,a3\,c\,x^5 + a3\,x^6$

Now try, notice that I have to indicate the variable in the command below:

```
In[15]:=  CoefficientList[Collect[Expand[{a0, a1, a2, a3}.{1, x, x^2, x^3} /. {x → (x + c)^2}], x], x]
```

Out[15]= $\{a0 + a1\,c^2 + a2\,c^4 + a3\,c^6,\ 2\,a1\,c + 4\,a2\,c^3 + 6\,a3\,c^5,\ a1 + 6\,a2\,c^2 + 15\,a3\,c^4,\ 4\,a2\,c + 20\,a3\,c^3,\ a2 + 15\,a3\,c^2,\ 6\,a3\,c,\ a3\}$

The next command makes sure that I have all the coefficients. Notice the difference

```
In[16]:=  CoefficientList[x + x^2, x]

Out[16]=  {0, 1, 1}

In[18]:=  PadRight[CoefficientList[x + x^2, x], 5]

Out[18]=  {0, 1, 1, 0, 0}
```

Notice the index **k** in **IdentityMatrix[5][[k]]**; this index selects the k-th row. Since this matrix has 5 rows I go through all of them by using **Table[ –stuff here – , {k, 1, 5}]**

As often in *Mathematica* I obtain rows what should be columns, so I transpose. I name the matrix mS and then show it easy to read using **MatrixForm[ ]**

This is the end of the explanation.

The above process can be automated for any degree:

```
In[19]:=  deg = 7

Out[19]=  7
```

To create the powers of x I use table, but I have to prepend 1

```
In[20]:=  Prepend[Table[x^j, {j, 1, deg}], 1]

Out[20]=  {1, x, x^2, x^3, x^4, x^5, x^6, x^7}
```

This is the only change in the above code.

```
In[24]:=  deg = 9; MatrixForm[mS = Transpose[Table[PadRight[CoefficientList[Collect[Expand[
            (IdentityMatrix[deg + 1][[k]].Prepend[Table[x^j, {j, 1, deg}], 1]) /. {x → (x + c)}
          ], x], x], deg + 1], {k, 1, deg + 1}]]]

Out[24]//MatrixForm=
```

$$\begin{pmatrix} 1 & c & c^2 & c^3 & c^4 & c^5 & c^6 & c^7 & c^8 & c^9 \\ 0 & 1 & 2c & 3c^2 & 4c^3 & 5c^4 & 6c^5 & 7c^6 & 8c^7 & 9c^8 \\ 0 & 0 & 1 & 3c & 6c^2 & 10c^3 & 15c^4 & 21c^5 & 28c^6 & 36c^7 \\ 0 & 0 & 0 & 1 & 4c & 10c^2 & 20c^3 & 35c^4 & 56c^5 & 84c^6 \\ 0 & 0 & 0 & 0 & 1 & 5c & 15c^2 & 35c^3 & 70c^4 & 126c^5 \\ 0 & 0 & 0 & 0 & 0 & 1 & 6c & 21c^2 & 56c^3 & 126c^4 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 7c & 28c^2 & 84c^3 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 8c & 36c^2 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 9c \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

---

# Legendre polynomials

Compared to the previous example, the only difference in the formula below is the definition of the transformation. In *Mathematica* we take derivative like this

```
In[74]:=  D[x^3 + x^5, x]

Out[74]=  3 x^2 + 5 x^4
```

and the second derivative is just

```
In[75]:=  D[x^3 + x^5, {x, 2}]

Out[75]=  6 x + 20 x^3
```

Here is the differential operator

```
In[71]:=  ((1 - x^2) D[#, {x, 2}] - 2 x D[#, x]) &[x^3]

Out[71]=  -6 x^3 + 6 x (1 - x^2)
```

or,

```
In[72]:=  ((1 - x^2) D[#, {x, 2}] - 2 x D[#, x]) &[a0 + a1 x + a2 x^2]

Out[72]=  -2 x (a1 + 2 a2 x) + 2 a2 (1 - x^2)
```

Notice that in the above command a polynomial comes at the end in square brackets;

& sign is just a part of the code;

the operator is defined inside of ( -differential operator - )&

the symbol # is replaced by the polynomial given at the end in [ ]

```
MatrixForm[mL1 = Transpose[Table[PadRight[CoefficientList[Collect[Expand[Simplify[
          ((1 - x^2) D[#, {x, 2}] - 2 x D[#, x]) &[
           (IdentityMatrix[5][[k]].{1, x, x^2, x^3, x^4})
          ]
         ]], x], x], 5], {k, 1, 5}]]]
```

$$\begin{pmatrix} 0 & 0 & 2 & 0 & 0 \\ 0 & -2 & 0 & 6 & 0 \\ 0 & 0 & -6 & 0 & 12 \\ 0 & 0 & 0 & -12 & 0 \\ 0 & 0 & 0 & 0 & -20 \end{pmatrix}$$

```
esL1 = Eigensystem[mL1]
```

```
{{-20, -12, -6, -2, 0},
 {{3, 0, -30, 0, 35}, {0, -3, 0, 5, 0}, {-1, 0, 3, 0, 0}, {0, 1, 0, 0, 0}, {1, 0, 0, 0, 0}}}
```

```
Reverse[({1, x, x^2, x^3, x^4}.#) & /@ esL1[[2]]]
```

$\{1, x, -1 + 3 x^2, -3 x + 5 x^3, 3 - 30 x^2 + 35 x^4\}$

Normalization is by $p(1) = 1$.

```
(# / (# /. {x → 1})) & /@ Reverse[({1, x, x^2, x^3, x^4}.#) & /@ esL1[[2]]]
```

$\left\{1, x, \frac{1}{2} (-1 + 3 x^2), \frac{1}{2} (-3 x + 5 x^3), \frac{1}{8} (3 - 30 x^2 + 35 x^4)\right\}$

```
Plot[{1, x, 1/2 (-1 + 3 x^2), 1/2 (-3 x + 5 x^3), 1/8 (3 - 30 x^2 + 35 x^4)},
 {x, -1.1, 1.1}, PlotStyle -> {{Red, Thickness[0.007]}, {Cyan, Thickness[0.007]},
   {Green, Thickness[0.007]}, {Magenta, Thickness[0.007]}, {Blue, Thickness[0.007]}},
 PlotRange -> {-1.2, 1.2}, ImageSize -> 400]
```




```
- Graphics -
```

```
Expand[Table[1/(2^n n!) D[(x^2 - 1)^n, {x, n}], {n, 1, 4}]]
```

$$\left\{x,\ -\frac{1}{2}+\frac{3x^2}{2},\ -\frac{3x}{2}+\frac{5x^3}{2},\ \frac{3}{8}-\frac{15x^2}{4}+\frac{35x^4}{8}\right\}$$

---

## Legendre polynomials (set your degree)

```
In[32]:= deg = 7;
```

```
In[33]:= Clear[mL1];
         MatrixForm[mL1 = Transpose[Table[PadRight[CoefficientList[Collect[Expand[Simplify[
                   ((1 - x^2) D[#, {x, 2}] - 2 x D[#, x]) &[
                    (IdentityMatrix[deg + 1][[k]].Prepend[Table[x^j, {j, 1, deg}], 1])
                   ]
                  ]], x], x], deg + 1], {k, 1, deg + 1}]]]
```

Out[33]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 2 & 0 & 0 & 0 & 0 & 0 \\ 0 & -2 & 0 & 6 & 0 & 0 & 0 & 0 \\ 0 & 0 & -6 & 0 & 12 & 0 & 0 & 0 \\ 0 & 0 & 0 & -12 & 0 & 20 & 0 & 0 \\ 0 & 0 & 0 & 0 & -20 & 0 & 30 & 0 \\ 0 & 0 & 0 & 0 & 0 & -30 & 0 & 42 \\ 0 & 0 & 0 & 0 & 0 & 0 & -42 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & -56 \end{pmatrix}$$

```
In[34]:= esL1 = Eigensystem[mL1]

Out[34]= {{-56, -42, -30, -20, -12, -6, -2, 0},
  {{0, -35, 0, 315, 0, -693, 0, 429}, {-5, 0, 105, 0, -315, 0, 231, 0},
   {0, 15, 0, -70, 0, 63, 0, 0}, {3, 0, -30, 0, 35, 0, 0, 0}, {0, -3, 0, 5, 0, 0, 0, 0},
   {-1, 0, 3, 0, 0, 0, 0, 0}, {0, 1, 0, 0, 0, 0, 0, 0}, {1, 0, 0, 0, 0, 0, 0, 0}}}
```

In[35]:= `Reverse[(Prepend[Table[x^j, {j, 1, deg}], 1].#) & /@ esL1[[2]]]`

Out[35]= $\{1, x, -1+3x^2, -3x+5x^3, 3-30x^2+35x^4, 15x-70x^3+63x^5, -5+105x^2-315x^4+231x^6, -35x+315x^3-693x^5+429x^7\}$

Normalization: $p(1) = 1$

In[36]:= `polysL1 =` $\left(\frac{\#}{\# \,/.\, \{x \to 1\}}\right)$ `& /@ Reverse[(Prepend[Table[x^j, {j, 1, deg}], 1].#) & /@ esL1[[2]]]`

Out[36]= $\left\{1, x, \frac{1}{2}(-1+3x^2), \frac{1}{2}(-3x+5x^3), \frac{1}{8}(3-30x^2+35x^4), \frac{1}{8}(15x-70x^3+63x^5), \frac{1}{16}(-5+105x^2-315x^4+231x^6), \frac{1}{16}(-35x+315x^3-693x^5+429x^7)\right\}$

```
In[37]:= Plot[Evaluate[polysL1], {x, -1.1, 1.1},
  PlotStyle -> {{Red, Thickness[0.007]}, {Cyan, Thickness[0.007]},
    {Green, Thickness[0.007]}, {Magenta, Thickness[0.007]}, {Blue, Thickness[0.007]},
    {Yellow, Thickness[0.007]}}, PlotRange -> {-1.5, 1.5}, ImageSize -> 400]
```




Out[37]= - Graphics -

In[31]:= `Expand[Table[`$\frac{1}{2^n\, n!}$` D[(x^2 - 1)^n, {x, n}], {n, 0, deg}]]`

Out[31]= $\left\{1, x, -\frac{1}{2}+\frac{3x^2}{2}, -\frac{3x}{2}+\frac{5x^3}{2}, \frac{3}{8}-\frac{15x^2}{4}+\frac{35x^4}{8}, \frac{15x}{8}-\frac{35x^3}{4}+\frac{63x^5}{8}\right\}$

## Laguerre polynomials

In[38]:=
```
deg = 7;
MatrixForm[mL2 = Transpose[Table[PadRight[CoefficientList[Collect[Expand[Simplify[
          (x D[#, {x, 2}] + (1 - x) D[#, x]) &[
           (IdentityMatrix[deg + 1][[k]].Prepend[Table[x^j, {j, 1, deg}], 1])
          ]
         ]], x], x], deg + 1], {k, 1, deg + 1}]]]
```

Out[38]//MatrixForm=

$$\begin{pmatrix} 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & -1 & 4 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -2 & 9 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -3 & 16 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -4 & 25 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -5 & 36 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & -6 & 49 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & -7 \end{pmatrix}$$

In[39]:=
```
esL2 = Eigensystem[mL2]
```

Out[39]=
```
{{-7, -6, -5, -4, -3, -2, -1, 0}, {{-5040, 35280, -52920, 29400, -7350, 882, -49, 1},
  {720, -4320, 5400, -2400, 450, -36, 1, 0}, {-120, 600, -600, 200, -25, 1, 0, 0},
  {24, -96, 72, -16, 1, 0, 0, 0}, {-6, 18, -9, 1, 0, 0, 0, 0},
  {2, -4, 1, 0, 0, 0, 0, 0}, {-1, 1, 0, 0, 0, 0, 0, 0}, {1, 0, 0, 0, 0, 0, 0, 0}}}
```

In[41]:=
```
Reverse[(Prepend[Table[x^j, {j, 1, deg}], 1].#) & /@ esL2[[2]]]
```

Out[41]=

$\{1, -1+x, 2-4x+x^2, -6+18x-9x^2+x^3, 24-96x+72x^2-16x^3+x^4,$
$-120+600x-600x^2+200x^3-25x^4+x^5, 720-4320x+5400x^2-2400x^3+450x^4-36x^5+x^6,$
$-5040+35280x-52920x^2+29400x^3-7350x^4+882x^5-49x^6+x^7\}$

Noramlization: $p(0) = 1$.

In[42]:=
```
polysL2 =
  (# / (# /. {x -> 0})) & /@ Reverse[(Prepend[Table[x^j, {j, 1, deg}], 1].#) & /@ esL2[[2]]]
```

Out[42]=

$\{1, 1-x, \frac{1}{2}(2-4x+x^2), \frac{1}{6}(6-18x+9x^2-x^3),$
$\frac{1}{24}(24-96x+72x^2-16x^3+x^4), \frac{1}{120}(120-600x+600x^2-200x^3+25x^4-x^5),$
$\frac{1}{720}(720-4320x+5400x^2-2400x^3+450x^4-36x^5+x^6), \frac{1}{5040}$
$(5040-35280x+52920x^2-29400x^3+7350x^4-882x^5+49x^6-x^7)\}$

```
In[43]:= Plot[Evaluate[polysL2], {x, -1, 5},
          PlotStyle → {{Red, Thickness[0.007]}, {Cyan, Thickness[0.007]},
            {Green, Thickness[0.007]}, {Magenta, Thickness[0.007]}, {Blue, Thickness[0.007]}},
          PlotRange → {-2, 4}, ImageSize → 400]
```




```
Out[43]= - Graphics -
```

# Bessel polynomials

```
In[44]:= deg = 7;
         MatrixForm[mB = Transpose[Table[PadRight[CoefficientList[Collect[Expand[Simplify[
                   (x^2 D[#, {x, 2}] + 2 (1 + x) D[#, x]) &[
                    (IdentityMatrix[deg + 1][[k]].Prepend[Table[x^j, {j, 1, deg}], 1])
                   ]
                 ]], x], x], deg + 1], {k, 1, deg + 1}]]]
```

Out[44]//MatrixForm=

$$\begin{pmatrix} 0 & 2 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 2 & 4 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 6 & 6 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 12 & 8 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 20 & 10 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 30 & 12 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 42 & 14 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 56 \end{pmatrix}$$

```
In[45]:= esB = Eigensystem[mB]

Out[45]= {{56, 42, 30, 20, 12, 6, 2, 0}, {{1, 28, 378, 3150, 17325, 62370, 135135, 135135},
           {1, 21, 210, 1260, 4725, 10395, 10395, 0}, {1, 15, 105, 420, 945, 945, 0, 0},
           {1, 10, 45, 105, 105, 0, 0, 0}, {1, 6, 15, 15, 0, 0, 0, 0},
           {1, 3, 3, 0, 0, 0, 0, 0}, {1, 1, 0, 0, 0, 0, 0, 0}, {1, 0, 0, 0, 0, 0, 0, 0}}}
```

```
In[46]:= Reverse[(Prepend[Table[x^j, {j, 1, deg}], 1].#) & /@ esB[[2]]]

Out[46]= {1, 1 + x, 1 + 3 x + 3 x^2, 1 + 6 x + 15 x^2 + 15 x^3, 1 + 10 x + 45 x^2 + 105 x^3 + 105 x^4,
  1 + 15 x + 105 x^2 + 420 x^3 + 945 x^4 + 945 x^5, 1 + 21 x + 210 x^2 + 1260 x^3 + 4725 x^4 + 10395 x^5 + 10395 x^6,
  1 + 28 x + 378 x^2 + 3150 x^3 + 17325 x^4 + 62370 x^5 + 135135 x^6 + 135135 x^7}
```

Normalization: $p(0) = 1$

```
In[47]:= polysB = (# / (# /. {x -> 0})) & /@ Reverse[(Prepend[Table[x^j, {j, 1, deg}], 1].#) & /@ esB[[2]]]

Out[47]= {1, 1 + x, 1 + 3 x + 3 x^2, 1 + 6 x + 15 x^2 + 15 x^3, 1 + 10 x + 45 x^2 + 105 x^3 + 105 x^4,
  1 + 15 x + 105 x^2 + 420 x^3 + 945 x^4 + 945 x^5, 1 + 21 x + 210 x^2 + 1260 x^3 + 4725 x^4 + 10395 x^5 + 10395 x^6,
  1 + 28 x + 378 x^2 + 3150 x^3 + 17325 x^4 + 62370 x^5 + 135135 x^6 + 135135 x^7}

In[48]:= Plot[Evaluate[polysB], {x, -1, 1},
  PlotStyle -> {{Red, Thickness[0.007]}, {Cyan, Thickness[0.007]},
    {Green, Thickness[0.007]}, {Magenta, Thickness[0.007]}, {Blue, Thickness[0.007]}},
  PlotRange -> {-2, 5}, ImageSize -> 400]
```




```
Out[48]= - Graphics -
```

## Chebyshev polynomials

```
deg = 5;
```

```
Clear[mCh];
MatrixForm[mCh = Transpose[Table[PadRight[CoefficientList[Collect[Expand[Simplify[
        ((1 - x^2) D[#, {x, 2}] - (x) D[#, x]) &[
         (IdentityMatrix[deg + 1][[k]].Prepend[Table[x^j, {j, 1, deg}], 1])
        ]
      ]], x], x], deg + 1], {k, 1, deg + 1}]]]
```

$$\begin{pmatrix} 0 & 0 & 2 & 0 & 0 & 0 \\ 0 & -1 & 0 & 6 & 0 & 0 \\ 0 & 0 & -4 & 0 & 12 & 0 \\ 0 & 0 & 0 & -9 & 0 & 20 \\ 0 & 0 & 0 & 0 & -16 & 0 \\ 0 & 0 & 0 & 0 & 0 & -25 \end{pmatrix}$$

```
esCh = Eigensystem[mCh]
```

```
{{-25, -16, -9, -4, -1, 0}, {{0, 5, 0, -20, 0, 16}, {1, 0, -8, 0, 8, 0},
  {0, -3, 0, 4, 0, 0}, {-1, 0, 2, 0, 0, 0}, {0, 1, 0, 0, 0, 0}, {1, 0, 0, 0, 0, 0}}}
```

```
Reverse[(Prepend[Table[x^j, {j, 1, deg}], 1].#) & /@ esCh[[2]]]
```

$\{1, x, -1 + 2x^2, -3x + 4x^3, 1 - 8x^2 + 8x^4, 5x - 20x^3 + 16x^5\}$

Normalization: $p(1) = 1$

```
polysCh = (# / (# /. {x -> 1})) & /@ Reverse[(Prepend[Table[x^j, {j, 1, deg}], 1].#) & /@ esCh[[2]]]
```

$\{1, x, -1 + 2x^2, -3x + 4x^3, 1 - 8x^2 + 8x^4, 5x - 20x^3 + 16x^5\}$

```
Plot[Evaluate[polysCh], {x, -1.1, 1.1},
 PlotStyle -> {{Red, Thickness[0.007]}, {Cyan, Thickness[0.007]},
   {Green, Thickness[0.007]}, {Magenta, Thickness[0.007]}, {Blue, Thickness[0.007]},
   {Yellow, Thickness[0.007]}}, PlotRange -> {-1.5, 1.5}, ImageSize -> 400]
```




- Graphics -

```
Table[TrigExpand[Cos[n ArcCos[x]]], {n, 0, 5}]
```

$$\{1, x, -1+2x^2, -3x+4x^3, 1-8x^2+8x^4, 5x-20x^3+16x^5\}$$

## Chebyshev polynomials - many

```
deg = 10;
```

```
Clear[mCh];
MatrixForm[mCh = Transpose[Table[PadRight[CoefficientList[Collect[Expand[Simplify[
        ((1 - x^2) D[#, {x, 2}] - (x) D[#, x]) &[
         (IdentityMatrix[deg + 1][[k]].Prepend[Table[x^j, {j, 1, deg}], 1])
        ]
       ]], x], x], deg + 1], {k, 1, deg + 1}]]]
```

$$\begin{pmatrix}
0 & 0 & 2 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & -1 & 0 & 6 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & -4 & 0 & 12 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & -9 & 0 & 20 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & -16 & 0 & 30 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & -25 & 0 & 42 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & -36 & 0 & 56 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & -49 & 0 & 72 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -64 & 0 & 90 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -81 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -100
\end{pmatrix}$$

```
esCh = Eigensystem[mCh]
```

```
{{-100, -81, -64, -49, -36, -25, -16, -9, -4, -1, 0},
 {{-1, 0, 50, 0, -400, 0, 1120, 0, -1280, 0, 512},
  {0, 9, 0, -120, 0, 432, 0, -576, 0, 256, 0}, {1, 0, -32, 0, 160, 0, -256, 0, 128, 0, 0},
  {0, -7, 0, 56, 0, -112, 0, 64, 0, 0, 0}, {-1, 0, 18, 0, -48, 0, 32, 0, 0, 0, 0},
  {0, 5, 0, -20, 0, 16, 0, 0, 0, 0, 0}, {1, 0, -8, 0, 8, 0, 0, 0, 0, 0, 0},
  {0, -3, 0, 4, 0, 0, 0, 0, 0, 0, 0}, {-1, 0, 2, 0, 0, 0, 0, 0, 0, 0, 0},
  {0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}}}
```

```
Reverse[(Prepend[Table[x^j, {j, 1, deg}], 1].#) & /@ esCh[[2]]]
```

$$\{1, x, -1+2x^2, -3x+4x^3, 1-8x^2+8x^4, 5x-20x^3+16x^5, -1+18x^2-48x^4+32x^6, -7x+56x^3-112x^5+64x^7, 1-32x^2+160x^4-256x^6+128x^8, 9x-120x^3+432x^5-576x^7+256x^9, -1+50x^2-400x^4+1120x^6-1280x^8+512x^{10}\}$$

Normalization: $p(1) = 1$

```
polysCh = (# / (# /. {x -> 1})) & /@ Reverse[(Prepend[Table[x^j, {j, 1, deg}], 1].#) & /@ esCh[[2]]]
```

$$\{1, x, -1+2x^2, -3x+4x^3, 1-8x^2+8x^4, 5x-20x^3+16x^5, -1+18x^2-48x^4+32x^6, -7x+56x^3-112x^5+64x^7, 1-32x^2+160x^4-256x^6+128x^8, 9x-120x^3+432x^5-576x^7+256x^9, -1+50x^2-400x^4+1120x^6-1280x^8+512x^{10}\}$$

```
Plot[Evaluate[polysCh], {x, -1.1, 1.1},
 PlotStyle → {{Red, Thickness[0.007]}, {Cyan, Thickness[0.007]},
   {Green, Thickness[0.007]}, {Magenta, Thickness[0.007]}, {Blue, Thickness[0.007]},
   {Yellow, Thickness[0.007]}}, PlotRange → {-1.5, 1.5}, ImageSize → 400]
```




```
- Graphics -
```

```
Table[TrigExpand[Cos[n ArcCos[x]]], {n, 0, deg}]
```

$\{1, x, -1+2x^2, -3x+4x^3, 1-8x^2+8x^4, 5x-20x^3+16x^5, -1+18x^2-48x^4+32x^6, -7x+56x^3-112x^5+64x^7, 1-32x^2+160x^4-256x^6+128x^8, 9x-120x^3+432x^5-576x^7+256x^9, -1+50x^2-400x^4+1120x^6-1280x^8+512x^{10}\}$

## ■ Looking for a general formula for Chebyshev polynomials

### ■ Even degree

```
nn = 5; ttt = Table[{(2 nn)^2 - (2 k)^2, (2 k + 2) (2 k + 1)}, {k, 0, nn - 1}]
```

```
{{100, 2}, {96, 12}, {84, 30}, {64, 56}, {36, 90}}
```

$\left(\frac{\#[[2]]}{\#[[1]]}\right)$ & /@ ttt

$\left\{\frac{1}{50}, \frac{1}{8}, \frac{5}{14}, \frac{7}{8}, \frac{5}{2}\right\}$

$\frac{5}{2}\,512$

1280

$\frac{7}{8}\,1280$

1120

```
2^(2 nn - 1) Table[(-1)^(j+1) Product[((2 k + 2) (2 k + 1))/((2 nn)^2 - (2 k)^2), {k, j, nn - 1}], {j, 0, nn}]
```

```
{-1, 50, -400, 1120, -1280, 512}
```

```
nn = 11;
```

```
2^(2 nn - 1) Table[(-1)^(j+1) Product[((2 k + 2) (2 k + 1))/((2 nn)^2 - (2 k)^2), {k, j, nn - 1}], {j, 0, nn}]
```

```
{-1, 242, -9680, 151008, -1208064, 5637632,
 -16400384, 30638080, -36765696, 27394048, -11534336, 2097152}
```

```
Total[%]
```

```
1
```

## Hermite polynomials

```
In[65]:= deg = 5;
         MatrixForm[mH = Transpose[Table[PadRight[CoefficientList[Collect[Expand[Simplify[
                   ((1) D[#, {x, 2}] - (2 x) D[#, x]) &[
                    (IdentityMatrix[deg + 1][[k]].Prepend[Table[x^j, {j, 1, deg}], 1])
                   ]
                 ]], x], x], deg + 1], {k, 1, deg + 1}]]]
```

Out[65]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 2 & 0 & 0 & 0 \\ 0 & -2 & 0 & 6 & 0 & 0 \\ 0 & 0 & -4 & 0 & 12 & 0 \\ 0 & 0 & 0 & -6 & 0 & 20 \\ 0 & 0 & 0 & 0 & -8 & 0 \\ 0 & 0 & 0 & 0 & 0 & -10 \end{pmatrix}$$

```
In[66]:= esH = Eigensystem[mH]
```

```
Out[66]= {{-10, -8, -6, -4, -2, 0}, {{0, 15, 0, -20, 0, 4}, {3, 0, -12, 0, 4, 0},
           {0, -3, 0, 2, 0, 0}, {-1, 0, 2, 0, 0, 0}, {0, 1, 0, 0, 0, 0}, {1, 0, 0, 0, 0, 0}}}
```

```
In[67]:= Reverse[(Prepend[Table[x^j, {j, 1, deg}], 1].#) & /@ esH[[2]]]
```

Out[67]= $\{1, x, -1 + 2 x^2, -3 x + 2 x^3, 3 - 12 x^2 + 4 x^4, 15 x - 20 x^3 + 4 x^5\}$

The normalization: the leading coefficient is $2^n$.

```
In[68]:= polysH = Expand[(2^(Length[CoefficientList[#,x]]-1) #/Last[CoefficientList[#, x]])] & /@
           Reverse[(Prepend[Table[x^j, {j, 1, deg}], 1].#) & /@ esH[[2]]]
```

Out[68]= $\{1, 2 x, -2 + 4 x^2, -12 x + 8 x^3, 12 - 48 x^2 + 16 x^4, 120 x - 160 x^3 + 32 x^5\}$

In[69]:=
```
Plot[Evaluate[polysH], {x, -3, 3},
 PlotStyle → {{Red, Thickness[0.007]}, {Cyan, Thickness[0.007]},
   {Green, Thickness[0.007]}, {Magenta, Thickness[0.007]}, {Blue, Thickness[0.007]},
   {Yellow, Thickness[0.007]}}, PlotRange → {-35, 40}, ImageSize → 400]
```




Out[69]= - Graphics -

In[70]:=
```
Expand[Table[Simplify[(-1)^n Exp[x^2] D[Exp[-x^2], {x, n}]], {n, 0, deg}]]
```

Out[70]= $\{1, 2x, -2+4x^2, -12x+8x^3, 12-48x^2+16x^4, 120x-160x^3+32x^5\}$